

Spojování textů (textových řetězců) v Excelu patří sice mezi triviální záležitosti, přesto věřím, že si v dnešním článku každý najde to své. Pojdme na to.

Nebudu zde dlouze popisovat něco, co lze vyjádřit snadno obrázkem.



### Spojování textových řetězců na listu

Řetězce spojujeme s pomocí ampersandu (znaku &), nikoliv znakem plus jako u čísel. Výsledkem spojení textového řetězce a čísla je také textový řetězec. Ve vzorcích se textový řetězec objevuje v uvozovkách a mezera je znak jako každý jiný (na listu si vyzkoušejte funkci ZNAK(32)). Odřádkování mají na starosti také znaky ASCII tabulky (zpravidla se používá ZNAK(13) spolu se ZNAK(10), někdy stačí i samotné užití ZNAK(10)). Pozn. Ručního zalomení prostého textu v buňce dosáhnete stiskem ALT+ENTER.

#### Tip

Víte, jak jednoduše zřetězíte obsah několika buněk? Ve volné buňce se vzorcem odkážete na danou oblast a ještě v režimu editace stisknete F9. Dojde k částečnému výpočtu (dosazení hodnot). Výsledek ořízněte o úvodní znak rovná se a složené závorky a máte hotovo.



#### Tip na zřetězení (dílčí výpočet s pomocí F9)

Výstup není úplně ideální. Kupříkladu pro Data / Ověření bychom se museli ještě zbavit uvozovek a naopak užití ve VBA a jeho Array vyžaduje jako oddělovač čárku. Ukážeme si později, jak na to s vlastní funkcí.

A jak se provádí zřetězení pod VBA?

```

1  Sub SpojovaniRetezcuA()
2
3  Dim strJmeno As String
4  Dim strPrijmeni As String
5  Dim strJmenoPrijmeni As String
6  Dim strDvaRadky As String
7  Dim strTabulator As String
8
9  strJmeno = "František"
10 strPrijmeni = "Koudelka"
11
12 'prosté spojení řetězců
13 'nepoužívejte znak + pro spojení
14 strJmenoPrijmeni = strJmeno & " " & strPrijmeni
15
16 MsgBox strJmenoPrijmeni
17
18 'odřádkování
19 'konstanta Visual Basicu představující dva znaky ASCII
20 '13 (CR, Carriage Return, "návrat vozíku")
21 'a 10 (LF, Line Feed, "posun o řádek")
22 'lze použít i konstanty vbLf nebo vbNewLine
23 strDvaRadky = "1. řádek" & vbCrLf & "2. řádek"
24
25 MsgBox strDvaRadky
26
27 'tabulátor, odřádkování
28 strTabulator = "Položka 1: " & vbTab & "nenalezena" _
29               & vbNewLine _
30               & "Položka 2: " & vbTab & 36
31
32 MsgBox strTabulator
33
34 End Sub

```



## Prosté spojování řetězců ve VBA

Ve VBA ještě zůstaneme a úlohu rozšíříme pro více textových vstupů.

1	<b>Sub</b> Techniky()
2	
3	Const cstrOddelovac <b>As String</b> = ", "
4	
5	<b>Dim</b> aRetezce()
6	<b>Dim</b> strTempA <b>As String</b>
7	<b>Dim</b> strTempB <b>As String</b>
8	<b>Dim</b> rngBunka <b>As Range</b>
9	
10	'a) Technika nabalování
11	
12	'pro každou buňku v oblasti...
13	<b>For Each</b> rngBunka <b>In Range</b> ("rngRetezce")
14	'nabalení řetězce a oddělovače
15	strTempA = strTempA & rngBunka.Text & cstrOddelovac
16	<b>Next</b> rngBunka
17	
18	'oříznutí posledního oddělovače
19	strTempA = Left(strTempA, Len(strTempA) - Len(cstrOddelovac))
20	
21	'b) Technika Join
22	
23	'překlopení jednosloupcové pojmenované oblasti buněk
24	'do jednorozměrného (vodorovného) pole
25	aRetezce = WorksheetFunction.Transpose(Range("rngRetezce"))
26	
27	'sloučení položek pole do řetězce
28	'mezi nimi volitelný oddělovač
29	strTempB = Join(aRetezce, cstrOddelovac)
30	
31	<b>End Sub</b>

Výše uvedenou proceduru odkrojujte. Technika nabalování řetězců je nejrozšířenější, ale nepatří mezi efektivní. Vyžaduje spoustu paměťově náročných operací a ve finále musíme řetězec ještě ořezávat. Práci s polem a funkcemi Join (a její sestřičkou Split) si vryjte do paměti. Bude se vám hodit v mnoha kulišárnách.

Nyní naše programové snažení zapracujeme do tří vlastních funkcí.

```

1  Public Function epfRETEZA(Oblast As Range, Optional Oddelovac As String = _
2  vbNullString) As String
3
4  Dim rngBunka As Range
5  Dim strTemp As String
6
7  'pro každou buňku v oblasti
8  For Each rngBunka In Oblast
9  'pokud není buňka prázdná...
10 If Not IsEmpty(rngBunka) Then
11     'nabalování řetězců
12     strTemp = strTemp & rngBunka.Text & Oddelovac
13 End If
14 Next rngBunka
15
16 'řetězec oříznut od posledního oddělovače
17 epfRETEZA = Left(strTemp, Len(strTemp) - Len(Oddelovac))
18
19 End Function
20
21 Public Function epfRETEZB(Oblast As Range, Optional Oddelovac As String = _
22 vbNullString, Optional Uvozovky As Boolean = False, Optional Docistit As _
23 Boolean = True) As String
24
25 Dim strTemp As String
26 Dim strUvozovky As String
27
28 With WorksheetFunction
29
30     'Transpose (funkce listu TRANSPOZICE)
31     'převod pole do řetězce položek
32     'do vodorovného jednorozměrného pole
33
34     'Join
35     'převod pole do řetězce položek
36     'mezi nimi volitelně oddělovac
37
38     strTemp = Join(.Transpose(Oblast), Oddelovac)
39
40     strUvozovky = Chr(34)
41
42     'vložit položky do uvozovek?
43     If Uvozovky = True Then
44         'ano
45         strTemp = strUvozovky & Replace(strTemp, Oddelovac, strUvozovky & _
46             Oddelovac & strUvozovky) & strUvozovky
47     End If
48
49     'odstranit nadbytečné mezery?
50     If Docistit = True Then
51         'ano
52         'Trim (funkce listu PROČISTIT)
53         'ořezání od nadbytečných mezer
54         strTemp = .Trim(strTemp)
55     End If
56
57 End With
58
59 epfRETEZB = strTemp
60
61 End Function
62
63 Public Function epfODSTAVEC(Oblast As Range) As String
64
65     'převod pole do řetězce položek
66     'do vodorovného jednorozměrného pole
67     'funkce listu TRANSPOZICE
68
69     'převod pole do řetězce položek
70     'mezi nimi mezera
71
72     'ořezání od nadbytečných mezer
73     'funkce listu PROČISTIT
74
75     With WorksheetFunction
76         epfODSTAVEC = .Trim(Join(.Transpose(Oblast), Chr(32)))
77     End With
78
79 End Function

```

První dvě funkce epfRETEZA a epfRETEZB se liší užitou technikou. Druhá z nich pak ještě nabízí možnost zbavení se nadbytečných mezer. Funkce epfODSTAVEC se snad hodí v případě, kdy odstavec zkopírovaný přes schránku se v Excelu rozkouskuje do jednotlivých buněk a vy je budete chtít znovu poskládat do bloku. Pozn. Tento článek se nevěnuje funkcím do detailu, přesto se sluší říci alespoň to, že jejich parametry deklarované jako Optional jsou volitelné.



Vlastní funkce pro spojování řetězců

Příloha

[excel\\_spojovani\\_retezcu.zip](#)